# Revolution of two curves around the y-axis

I've looked through the other posts on rotating a function around the x or y axis and haven't had any luck.

my question is this how do i rotate the area between the functions $f(x)=x^2+1$, and $f(x)=2x^2-2$, whereby $x>0$ and $y>0$ and the region is bounded by the y and x axis.

This is a plot of the area I'm talking about.

Revolving the area around the y-axis is suppose to produce a vase, although I picture something of a shallow bowl.

Any help would be much appreciated. As I'm still quite new to Mathematica, an explanation of any code posted would be helpful.

• Welcome to Mathematica.SE! I suggest that: 1) You take the introductory Tour now! 2) When you see good questions and answers, vote them up by clicking the gray triangles, because the credibility of the system is based on the reputation gained by users sharing their knowledge. Also, please remember to accept the answer, if any, that solves your problem, by clicking the checkmark sign! 3) As you receive help, try to give it too, by answering questions in your area of expertise. – bbgodfrey May 8 '15 at 0:29
• Please include the code you are trying in your question – bbgodfrey May 8 '15 at 0:30
• RevolutionPlot3D[{x^2 + 1, 2 x^2 - 2}, {x, 0, Sqrt[3]}, PlotRange -> All, RevolutionAxis -> "Y"] this is the code I'm trying but its not giving me what i know it should look like – errymerry May 8 '15 at 0:33
• RevolutionPlot3D[{fx,fz},{t,t0,t1}] generates a plot of the surface obtained by rotating the parametric curve with x,z coordinates {fx, fz} around the z axis. So your input created a very different curve from what Plot shows. Check the documentation very carefully. – LLlAMnYP May 8 '15 at 14:32

I think the plot will look more like what I believe the OP is expecting if a few options are added.

Show[RevolutionPlot3D[#, {x, 0, Sqrt[3]},
PlotRange -> {Automatic, Automatic, {0., Automatic}},
MeshStyle -> None,
Axes -> None,
Boxed -> False,
PlotStyle -> Opacity[.5]] & /@
{2 x^2 - 2, x^2 + 1, Piecewise[{{0, 0 < x < 1}}, Null]}]


Also note that I have put a bottom on the bowl.

Show[RevolutionPlot3D[#, {x, 0, Sqrt[3]}, MeshStyle -> None,
PlotStyle -> Opacity[.5]] & /@ {2 x^2 - 2,  x^2 + 1}]


• shouldn't it have a flat bottom as the domain of this pair of functions is restricted to x>0 and y>0 ? – errymerry May 8 '15 at 1:52
• @errymerry I believe you can work that out easily by yourself ;) – Dr. belisarius May 8 '15 at 3:34

Just for something different:

pp = ParametricPlot3D[{Sqrt[3] Cos[t], Sqrt[3] Sin[t], 4}, {t, 0,
2 Pi}, PlotStyle -> {Red, Thickness[0.04]}, Boxed -> False];
rp = RegionPlot3D[
0 < x^2 + y^2 < 3 &&
Max[0, 2 x^2 + 2 y^2 - 2] < z < x^2 + y^2 + 1, {x, -2, 2}, {y, -2,
2}, {z, 0, 5}, Mesh -> None, PlotStyle -> Red, PlotPoints -> 200,
PerformanceGoal -> "Quality", Boxed -> False];
Show[pp, rp, Axes -> False, Background -> Black]


Top ring just to deal with boundary of intersection...